


## 別紙2 アセンブラ言語の仕様

### 1. システム COMET II の仕様

#### 1.1 ハードウェアの仕様

- (1) 1語は16ビットで、そのビット構成は、次のとおりである。



- (2) 主記憶の容量は65536語で、そのアドレスは0～65535番地である。  
 (3) 数値は、16ビットの2進数で表現する。負数は、2の補数で表現する。  
 (4) 制御方式は逐次制御で、命令語は1語長又は2語長である。  
 (5) レジスタとして、GR (16ビット)、SP (16ビット)、PR (16ビット)、FR (3ビット) の4種類がある。

GR (汎用レジスタ, General Register) は、GR0～GR7 の8個があり、算術、論理、比較、シフトなどの演算に用いる。このうち、GR1～GR7 のレジスタは、指標レジスタ (index register) としてアドレスの修飾にも用いる。

SP (スタックポインタ, Stack Pointer) は、スタックの最上段のアドレスを保持している。

PR (プログラムレジスタ, Program Register) は、次に実行すべき命令語の先頭アドレスを保持している。

FR (フラグレジスタ, Flag Register) は、OF (Overflow Flag)、SF (Sign Flag)、ZF (Zero Flag) と呼ぶ3個のビットからなり、演算命令などの実行によって次の値が設定される。これらの値は、条件付き分岐命令で参照される。

OF	算術演算命令の場合は、演算結果が-32768～32767に収まらなくなったとき1になり、それ以外るとき0になる。論理演算命令の場合は、演算結果が0～65535に収まらなくなったとき1になり、それ以外るとき0になる。
SF	演算結果の符号が負 (ビット番号15が1) のとき1、それ以外るとき0になる。
ZF	演算結果が零 (全部のビットが0) のとき1、それ以外るとき0になる。

- (6) 論理加算又は論理減算は、被演算データを符号のない数値とみなして、加算又は減算する。

#### 1.2 命令

命令の形式及びその機能を示す。ここで、一つの命令コードに対し2種類のオペランドがある場合、上段はレジスタ間の命令、下段はレジスタと主記憶間の命令を表す。

命 令	書 き 方		命 令 の 説 明	FRの設定
	命 令 コード	オペランド		

- (1) ロード、ストア、ロードアドレス命令

ロード Load	LD	$r1, r2$	$r1 \leftarrow (r2)$	○*1
		$r, \text{adr} [, x]$	$r \leftarrow (\text{実効アドレス})$	
ストア STore	ST	$r, \text{adr} [, x]$	実効アドレス $\leftarrow (r)$	—
ロードアドレス Load Address	LAD	$r, \text{adr} [, x]$	$r \leftarrow \text{実効アドレス}$	

## (2) 算術，論理演算命令

算術加算 ADD Arithmetic	ADDA	$r1, r2$ $r, \text{adr} [, x]$	$r1 \leftarrow (r1) + (r2)$ $r \leftarrow (r) + (\text{実効アドレス})$	○
論理加算 ADD Logical	ADDL	$r1, r2$ $r, \text{adr} [, x]$	$r1 \leftarrow (r1) +_L (r2)$ $r \leftarrow (r) +_L (\text{実効アドレス})$	
算術減算 SUBtract Arithmetic	SUBA	$r1, r2$ $r, \text{adr} [, x]$	$r1 \leftarrow (r1) - (r2)$ $r \leftarrow (r) - (\text{実効アドレス})$	
論理減算 SUBtract Logical	SUBL	$r1, r2$ $r, \text{adr} [, x]$	$r1 \leftarrow (r1) -_L (r2)$ $r \leftarrow (r) -_L (\text{実効アドレス})$	
論理積 AND	AND	$r1, r2$ $r, \text{adr} [, x]$	$r1 \leftarrow (r1) \text{ AND } (r2)$ $r \leftarrow (r) \text{ AND } (\text{実効アドレス})$	
論理和 OR	OR	$r1, r2$ $r, \text{adr} [, x]$	$r1 \leftarrow (r1) \text{ OR } (r2)$ $r \leftarrow (r) \text{ OR } (\text{実効アドレス})$	
排他的論理和 eXclusive OR	XOR	$r1, r2$ $r, \text{adr} [, x]$	$r1 \leftarrow (r1) \text{ XOR } (r2)$ $r \leftarrow (r) \text{ XOR } (\text{実効アドレス})$	○*1

## (3) 比較演算命令

(c) 比較演算命令																						
算術比較 ComPare Arithmetic	CPA	$r1, r2$	(r1) と (r2) , 又は (r) と (実効アドレス) の算術比較又は論理比較を行い, 比較結果によって, FR に次の値を設定する。				○*1															
		$r, \text{adr} [, x]$	<table><tr><th rowspan="2">比較結果</th><th colspan="2">FR の値</th></tr><tr><th>SF</th><th>ZF</th></tr><tr><td><math>(r1) &gt; (r2)</math></td><td rowspan="2">0</td><td rowspan="2">0</td></tr><tr><td><math>(r) &gt; (\text{実効アドレス})</math></td></tr><tr><td><math>(r1) = (r2)</math></td><td rowspan="2">0</td><td rowspan="2">1</td></tr><tr><td><math>(r) = (\text{実効アドレス})</math></td></tr><tr><td><math>(r1) &lt; (r2)</math></td><td rowspan="2">1</td><td rowspan="2">0</td></tr><tr><td><math>(r) &lt; (\text{実効アドレス})</math></td></tr></table>					比較結果	FR の値		SF	ZF	$(r1) > (r2)$	0	0	$(r) > (\text{実効アドレス})$	$(r1) = (r2)$	0	1	$(r) = (\text{実効アドレス})$	$(r1) < (r2)$	1
比較結果	FR の値																					
	SF	ZF																				
$(r1) > (r2)$	0	0																				
$(r) > (\text{実効アドレス})$																						
$(r1) = (r2)$	0	1																				
$(r) = (\text{実効アドレス})$																						
$(r1) < (r2)$	1	0																				
$(r) < (\text{実効アドレス})$																						
論理比較 ComPare Logical	CPL	$r1, r2$																				
		$r, \text{adr} [, x]$																				
		$r1, r2$																				
		$r, \text{adr} [, x]$																				
		$r1, r2$																				
		$r, \text{adr} [, x]$																				

## (4) シフト演算命令

算術左シフト Shift Left Arithmetic	SLA	$r, \text{adr} [, x]$	符号を除き (r) を実効アドレスで指定したビット数だけ左又は右にシフトする。	○*2
算術右シフト Shift Right Arithmetic	SRA	$r, \text{adr} [, x]$	シフトの結果，空いたビット位置には，左シフトのときは 0，右シフトのときは符号と同じものが入る。	
論理左シフト Shift Left Logical	SLL	$r, \text{adr} [, x]$	符号を含み (r) を実効アドレスで指定したビット数だけ左又は右にシフトする。	
論理右シフト Shift Right Logical	SRL	$r, \text{adr} [, x]$	シフトの結果，空いたビット位置には 0 が入る。	

## (5) 分岐命令

正分岐 Jump on Plus	JPL	adr [, x]	<div>FR の値によって、実効アドレスに分岐する。分岐しないときは、次の命令に進む。</div> <table><tr><th rowspan="2">命令</th><th colspan="3">分岐するときの FR の値</th></tr><tr><th>OF</th><th>SF</th><th>ZF</th></tr><tr><td>JPL</td><td></td><td>0</td><td>0</td></tr><tr><td>JMI</td><td></td><td>1</td><td></td></tr><tr><td>JNZ</td><td></td><td></td><td>0</td></tr><tr><td>JZE</td><td></td><td></td><td>1</td></tr><tr><td>JOV</td><td>1</td><td></td><td></td></tr></table>	命令	分岐するときの FR の値			OF	SF	ZF	JPL		0	0	JMI		1		JNZ			0	JZE			1	JOV	1		
命令	分岐するときの FR の値																													
	OF	SF		ZF																										
JPL		0		0																										
JMI		1																												
JNZ				0																										
JZE				1																										
JOV	1																													
負分岐 Jump on Minus	JMI	adr [, x]																												
非零分岐 Jump on Non Zero	JNZ	adr [, x]																												
零分岐 Jump on Zero	JZE	adr [, x]																												
オーバフロー分岐 Jump on Overflow	JOV	adr [, x]																												
無条件分岐 unconditional JUMP	JUMP	adr [, x]	無条件に実効アドレスに分岐する。																											

(6) スタック操作命令

プッシュ PUSH	PUSH    adr [,x]	SP ← (SP) − <sub>L</sub> 1, (SP) ← 実効アドレス	—
ポップ POP	POP       r	r ← ( (SP) ), SP ← (SP) + <sub>L</sub> 1	

(7) コール, リターン命令

コール CALL subroutine	CALL    adr [,x]	SP ← (SP) − <sub>L</sub> 1, (SP) ← (PR), PR ← 実効アドレス	—
リターン RETurn from subroutine	RET	PR ← ( (SP) ), SP ← (SP) + <sub>L</sub> 1	

(8) その他

スーパーバイザコール SuperVisor Call	SVC    adr [,x]	実効アドレスを引数として割出しを行 う。実行後の GR と FR は不定となる。	—
ノーオペレーション No OPeration	NOP	何もしない。	

注記    r, r1, r2    いずれも GR を示す。指定できる GR は GR0 ～ GR7  
           adr        アドレスを示す。指定できる値の範囲は 0 ～ 65535  
           x        指標レジスタとして用いる GR を示す。指定できる GR は GR1 ～ GR7  
           [     ]    [     ] 内の指定は省略できることを示す。  
           (     )    (     ) 内のレジスタ又はアドレスに格納されている内容を示す。  
           実効アドレス    adr と x の内容との論理加算値又はその値が示す番地  
           ←        演算結果を、左辺のレジスタ又はアドレスに格納することを示す。  
           +<sub>L</sub>, −<sub>L</sub>    論理加算, 論理減算を示す。  
           FR の設定    ○    : 設定されることを示す。  
                           ○\*1 : 設定されることを示す。ただし、OF には 0 が設定される。  
                           ○\*2 : 設定されることを示す。ただし、OF にはレジスタから最後に送り出  
                                   されたビットの値が設定される。  
                           —    : 実行前の値が保持されることを示す。

1.3 文字の符号表

- (1) JIS X 0201 ラテン文字・片仮名用 8 ビット符号  
で規定する文字の符号表を使用する。
- (2) 右に符号表の一部を示す。1 文字は 8 ビットか  
らなり、上位 4 ビットを列で、下位 4 ビットを行  
で示す。例えば、間隔, 4, H, ¥ のビット構成は、  
16 進表示で、それぞれ 20, 34, 48, 5C である。  
16 進表示で、ビット構成が 21 ～ 7E (及び表では  
省略している A1 ～ DF) に対応する文字を図形  
文字という。図形文字は、表示 (印刷) 装置で、  
文字として表示 (印字) できる。
- (3) この表にない文字とそのビット構成が必要な場  
合は、問題中で与える。

行 \ 列	02	03	04	05	06	07
0	間隔	0	@	P	`	p
1	!	1	A	Q	a	q
2	”	2	B	R	b	r
3	#	3	C	S	c	s
4	\$	4	D	T	d	t
5	%	5	E	U	e	u
6	&	6	F	V	f	v
7	’	7	G	W	g	w
8	(	8	H	X	h	x
9	)	9	I	Y	i	y
10	*	:	J	Z	j	z
11	+	;	K	L	k	{
12	,	<	L	¥	l	
13	-	=	M	]	m	}
14	.	>	N	^	n	~
15	/	?	0	_	o	

## 2. アセンブラ言語 CASL II の仕様

### 2.1 言語の仕様

- (1) CASL II は、COMET II のためのアセンブラ言語である。
- (2) プログラムは、命令行及び注釈行からなる。
- (3) 1 命令は 1 命令行で記述し、次の行へ継続できない。
- (4) 命令行及び注釈行は、次に示す記述の形式で、行の 1 文字目から記述する。

行 の 種 類		記 述 の 形 式
命令行	オペランドあり	[ラベル] {空白} {命令コード} {空白} {オペランド} [{空白} [コメント]]
	オペランドなし	[ラベル] {空白} {命令コード} [{空白} [{;} [コメント]]]
注釈行		[空白] {;} [コメント]

注記 [ ] [ ] 内の指定が省略できることを示す。

{ } { } 内の指定が必須であることを示す。

ラベル その命令の（先頭の語の）アドレスを他の命令やプログラムから参照するための名前である。長さは 1～8 文字で、先頭の文字は英大文字でなければならない。以降の文字は、英大文字又は数字のいずれでもよい。なお、予約語である GR0～GR7 は、使用できない。

空白 1 文字以上の間隔文字の列である。

命令コード 命令ごとに記述の形式が定義されている。

オペランド 命令ごとに記述の形式が定義されている。

コメント 覚え書きなどの任意の情報であり、処理系で許す任意の文字を書くことができる。

### 2.2 命令の種類

命令は、4 種類のアセンブラ命令（START, END, DS, DC）、4 種類のマクロ命令（IN, OUT, RPUSH, RPOP）及び機械語命令（COMET II の命令）からなる。その仕様を次に示す。

命令の種類	ラベル	命 令 コ ー ド	オペランド	機 能
アセンブラ命令	ラベル	START	[実行開始番地]	プログラムの先頭を定義 プログラムの実行開始番地を定義 他のプログラムで参照する入口名を定義
		END		プログラムの終わりを明示
	[ラベル]	DS	語数	領域を確保
	[ラベル]	DC	定数 [, 定数] …	定数を定義
マクロ命令	[ラベル]	IN	入力領域, 入力文字長領域	入力装置から文字データを入力
	[ラベル]	OUT	出力領域, 出力文字長領域	出力装置へ文字データを出力
	[ラベル]	RPUSH		GR の内容をスタックに格納
	[ラベル]	RPOP		スタックの内容を GR に格納
機械語命令	[ラベル]	(「1.2 命令」を参照)		

### 2.3 アセンブラ命令

アセンブラ命令は、アセンブラの制御などを行う。

- (1)

START 命令は、プログラムの先頭を定義する。

実行開始番地は、そのプログラム内で定義されたラベルで指定する。指定がある場合はその番地から、省略した場合は START 命令の次の命令から、実行を開始する。

また、この命令につけられたラベルは、他のプログラムから入口名として参照できる。

(2) 

END	
-----	--

END 命令は、プログラムの終わりを定義する。

(3) 

DS	語数
----	----

DS 命令は、指定した語数の領域を確保する。

語数は、10 進定数 ( $\geq 0$ ) で指定する。語数を 0 とした場合、領域は確保しないが、ラベルは有効である。

(4) 

DC	定数 [, 定数] ...
----	---------------

DC 命令は、定数で指定したデータを（連続する）語に格納する。

定数には、10 進定数、16 進定数、文字定数、アドレス定数の 4 種類がある。

定数の種類	書き方	命 令 の 説 明
10 進定数	n	n で指定した 10 進数値を、1 語の 2 進数データとして格納する。ただし、n が $-32768 \sim 32767$ の範囲にないときは、その下位 16 ビットを格納する。
16 進定数	#h	h は 4 けたの 16 進数（16 進数字は 0～9, A～F）とする。h で指定した 16 進数値を 1 語の 2 進数データとして格納する ( $0000 \leq h \leq FFFF$ )。
文字定数	'文字列'	文字列の文字数 ( $> 0$ ) 分の連続する領域を確保し、最初の文字は第 1 語の下位 8 ビットに、2 番目の文字は第 2 語の下位 8 ビットに、…と順次文字データとして格納する。各語の上位 8 ビットには 0 のビットが入る。 文字列には、間隔及び任意の図形文字を書くことができる。ただし、アポストロフィ (') は 2 個続けて書く。
アドレス定数	ラベル	ラベルに対応するアドレスを 1 語の 2 進数データとして格納する。

## 2.4 マクロ命令

マクロ命令は、あらかじめ定義された命令群とオペランドの情報によって、目的の機能を果たす命令群を生成する（語数は不定）。

(1) 

IN	入力領域, 入力文字長領域
----	---------------

IN 命令は、あらかじめ割り当てた入力装置から、1 レコードの文字データを読み込む。

入力領域は、256 語長の作業域のラベルであり、この領域の先頭から、1 文字を 1 語に対応させて順次入力される。レコードの区切り符号（キーボード入力の復帰符号など）は、格納しない。格納の形式は、DC 命令の文字定数と同じである。入力データが 256 文字に満たない場合、入力領域の残りの部分は実行前のデータを保持する。入力データが 256 文字を超える場合、以降の文字は無視される。

入力文字長領域は、1 語長の領域のラベルであり、入力された文字の長さ ( $\geq 0$ ) が 2 進数で格納される。ファイルの終わり (end of file) を検出した場合は、-1 が格納される。

IN 命令を実行すると、GR の内容は保存されるが、FR の内容は不定となる。

(2) 

OUT	出力領域, 出力文字長領域
-----	---------------

OUT 命令は、あらかじめ割り当てた出力装置に、文字データを、1 レコードとして書き出す。

出力領域は、出力しようとするデータが 1 文字 1 語で格納されている領域のラベルである。格納の形式は、DC 命令の文字定数と同じであるが、上位 8 ビットは、OS が無視するので 0 でなくてもよい。

出力文字長領域は、1 語長の領域のラベルであり、出力しようとする文字の長さ ( $\geq 0$ ) を 2 進数で格納しておく。

OUT 命令を実行すると、GR の内容は保存されるが、FR の内容は不定となる。

(3) 

R PUSH	
--------	--

R PUSH 命令は、GR の内容を、GR1, GR2, …, GR7 の順序でスタックに格納する。

(4) 

R POP	
-------	--

R POP 命令は、スタックの内容を順次取り出し、GR7, GR6, …, GR1 の順序で GR に格納する。

## 2.5 機械語命令

機械語命令のオペランドは、次の形式で記述する。

- r, r1, r2      GR は、記号 GR0 ~ GR7 で指定する。
- x                指標レジスタとして用いる GR は、記号 GR1 ~ GR7 で指定する。
- adr             アドレスは、10 進定数、16 進定数、アドレス定数又はリテラルで指定する。  
リテラルは、一つの 10 進定数、16 進定数又は文字定数の前に等号 (=) を付けて記述する。CASL II は、等号の後の定数をオペランドとする DC 命令を生成し、そのアドレスを adr の値とする。

## 2.6 その他

- (1) アセンブラによって生成される命令語や領域の相対位置は、アセンブラ言語での記述順序とする。ただし、リテラルから生成される DC 命令は、END 命令の直前にまとめて配置される。
- (2) 生成された命令語、領域は、主記憶上で連続した領域を占める。

## 3. プログラム実行の手引

### 3.1 OS

プログラムの実行に関して、次の取決めがある。

- (1) アセンブラは、未定義ラベル（オペランド欄に記述されたラベルのうち、そのプログラム内で定義されていないラベル）を、他のプログラムの入口名（START 命令のラベル）と解釈する。この場合、アセンブラはアドレスの決定を保留し、その決定を OS に任せる。OS は、実行に先立って他のプログラムの入口名との関係処理を行いアドレスを決定する（プログラムの関係）。
- (2) プログラムは、OS によって起動される。プログラムがロードされる主記憶の領域は不定とするが、プログラム中のラベルに対応するアドレス値は、OS によって実アドレスに補正されるものとする。
- (3) プログラムの起動時に、OS はプログラム用に十分な容量のスタック領域を確保し、その最後のアドレスに 1 を加算した値を SP に設定する。
- (4) OS は、CALL 命令でプログラムに制御を渡す。プログラムを終了し OS に制御を戻すときは、RET 命令を使用する。
- (5) IN 命令に対応する入力装置、OUT 命令に対応する出力装置の割当ては、プログラムの実行に先立って利用者が行う。
- (6) OS は、入出力装置や媒体による入出力手続の違いを吸収し、システムでの標準の形式及び手続（異常処理を含む）で入出力を行う。したがって、IN, OUT 命令では、入出力装置の違いを意識する必要はない。

### 3.2 未定義事項

プログラムの実行等に関し、この仕様で定義しない事項は、処理系によるものとする。

## 参考資料

参考資料は、COMET II の理解を助けるため又は COMET II の処理系作成者に対する便宜のための資料である。したがって、COMET II、CASL II の仕様に影響を与えるものではない。

### 1. 命令語の構成

命令語の構成は定義しないが、次のような構成を想定する。ここで、OP の数値は 16 進表示で示す。

15 11 7 3 0 15				0 ← ビット番号		
第 1 語				第 2 語	命令	命令語とアセンブラとの対応
OP	副OP	r/r1	x/r2	adr	語長	機械語命令
0	0	—	—	—	1	NOP
1	0				2	LD r, adr, x
	1				2	ST r, adr, x
	2				2	LAD r, adr, x
	4			—	1	LD r1, r2
2	0				2	ADDA r, adr, x
	1				2	SUBA r, adr, x
	2				2	ADDL r, adr, x
	3				2	SUBL r, adr, x
	4			—	1	ADDA r1, r2
	5			—	1	SUBA r1, r2
	6			—	1	ADDL r1, r2
3	0				2	AND r, adr, x
	1				2	OR r, adr, x
	2				2	XOR r, adr, x
	4			—	1	AND r1, r2
	5			—	1	OR r1, r2
	6			—	1	XOR r1, r2
4	0				2	CPA r, adr, x
	1				2	CPL r, adr, x
	4			—	1	CPA r1, r2
	5			—	1	CPL r1, r2
5	0				2	SLA r, adr, x
	1				2	SRA r, adr, x
	2				2	SLL r, adr, x
	3				2	SRL r, adr, x
6	1	—			2	JMI adr, x
	2	—			2	JNZ adr, x
	3	—			2	JZE adr, x
	4	—			2	JUMP adr, x
	5	—			2	JPL adr, x
	6	—			2	JOV adr, x
7	0	—			2	PUSH adr, x
	1		—	—	1	POP r
8	0	—			2	CALL adr, x
	1	—	—	—	1	RET
9 ～ E						その他の命令
F	0	—			2	SVC adr, x

## 2. マクロ命令

マクロ命令が生成する命令群は定義しない（語数不定）が，次の例のような命令群を生成することを想定する。


〔例〕 IN 命令

LABEL	IN	IBUF, LEN
	↓ マクロ生成	
LABEL	PUSH	0, GR1
	PUSH	0, GR2
	LAD	GR1, IBUF
	LAD	GR2, LEN
	SVC	1
	POP	GR2
	POP	GR1


## 3. シフト演算命令におけるビットの動き

シフト演算命令において，例えば，1 ビットのシフトをしたときの動き及び OF の変化は，次のとおりである。


- (1) 算術左シフトでは，ビット番号 14 の値が設定される。




- (2) 算術右シフトでは，ビット番号 0 の値が設定される。



- (3) 論理左シフトでは，ビット番号 15 の値が設定される。



- (4) 論理右シフトでは，ビット番号 0 の値が設定される。



## 4. プログラムの例

COUNT1	START		;
;	入力	GR1:検索する語	
;	処理	GR1 中の'1'のビットの個数を求める	
;	出力	GR0:GR1 中の'1'のビットの個数	
	PUSH	0, GR1	;
	PUSH	0, GR2	;
	SUBA	GR2, GR2	; Count = 0
	AND	GR1, GR1	; 全部のビットが'0'?
	JZE	RETURN	; 全部のビットが'0'なら終了
MORE	LAD	GR2, 1, GR2	; Count = Count + 1
	LAD	GR0, -1, GR1	; 最下位の'1'のビット1個を
	AND	GR1, GR0	; '0'に変える
	JNZ	MORE	; '1'のビットが残っていれば繰返し
RETURN	LD	GR0, GR2	; GR0 = Count
	POP	GR2	;
	POP	GR1	;
	RET		; 呼出しプログラムへ戻る
	END		;



Ver. 4.3 : 2021 年 10 月

■試験で使用する情報技術に関する用語・プログラム言語など■



独立行政法人情報処理推進機構  
Information-technology Promotion Agency, Japan

〒113-8663 東京都文京区本駒込 2-28-8  
文京グリーンコートセンターオフィス 15 階  
TEL 03-5978-7600 (代表)  
FAX 03-5978-7610



詳しくは…

<https://www.jitec.ipa.go.jp>